
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 7.1.4.2 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.4.2
Test Group:
\SDAP\7_1_4\
ATS Version:
iwd-TTCN3-B2019-12_D19wk38
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for SDAP test case 7.1.4.2 on wk38 ATS

Note : Changes 1 to 5 from TTCN CR R5s190946 and change 2 from TTCN CR R5s190944 are also required and applied.
Change 1
	Function name
	f_TC_7_1_4_2_NR5GC

	Reason for change
	1. The PDU session ID used in SDAP configuration should reflect the value received from the UE IN PDU SESSION ESTABLISHMENT REQUEST
2. The function f_NR_GetNextDRBIds() may not return DRB IDs 1 and 2 which are mandated by the tets case prose.
3. SS does not know the if UE is configured to include UL SDAP header
4. Test Loop is not deactivated  before postamble



	Summary of change
	1. Use PDU session ID from the established PDU session instead of ‘1’
Note : An associated prose CR on 38.523-1 will be raised atRAN5#85

2. Set DRB IDs to 1 and 2 as per test case prose.
3. Configure the presence of UL header for test mode (see changes 1,2 and 4)
4. Deactivate test loop after test case body


	TTCN module
	SDAP_NR5GC

	MCC160 Comment
	1. Accepted, conditional to prose CR approval.

2. Rejected.This is not the understanding of MCC 160 of the spec. We dont think the prose intends to hardcode DRB-Id's to 1,2,3 for both SA and NSA. The DRB ID as IE value is not specified in test case specific parameters.
3. Accepted.

4. Accepted.


Before change

        function f_TC_7_1_4_2_NR5GC() runs on NR5GC_PTC

  {//SDAP Data Transfer handling without Header UL/DL

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;

    var IntegerList_Type v_DRBList;

    var integer v_FirstDRB;

    var integer v_SecondDRB;

    var template (value) QoS_Rules v_QoS_Rules;

    var template (value) QoSFlowDescr v_QoSFlowDescr;

    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var PDN_Index_Type v_PDN_Index;

    var template (value) QosFlow_Identification_Type v_QosFlowId;

    f_NR5GC_Init_NAS(NR_1);

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    //Preamble

    // Perform Registration and get PDU Establishment Request message

    v_GSM_MobilityInfo_Type := fl_SDAP_Preamble_Part1 (nr_Cell1);

    //Get DRB identities

    v_DRBList:= f_NR_GetNextDRBIds(v_GSM_MobilityInfo_Type, true); // First is Default DRB

    v_FirstDRB := v_DRBList[0];

    v_SecondDRB := v_DRBList[1];
    //Note: In order to be able to start the IP handling during the preamble, QFI=2 is mapped temporarily to the default DRB (v_FirtDRB).

    //      This mapping will be removed when moving to connected mode

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_SecondDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_Configuration(1, Absent, Absent)), //Configure SDAP in normal mode for the moment

                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_FirstDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_Configuration(1, Absent, Present, {2}))}; //Configure SDAP in normal mode for the moment

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(1, absent, absent, false),

                                                   v_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(1, absent, present_, true, {2}),

                                                   v_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_QoS_Rules := cs_QoS_Rules ({fl_BuildQoSRuleRemoteAccess("4"), fl_BuildQoSRuleRemoteAccess("5")});

    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow2, cs_QoS_Flow3});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,

    // then stop it before the test body, so that TTCN can route the IP data

    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);

    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId.sessionId), 2); //Indicate the QoS flow associated with the default DRB to start the IP handling

    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));

    fl_SDAP_Preamble_Part2 (nr_Cell1,

                            v_GSM_MobilityInfo_Type,

                            v_FirstDRB,

                            v_SecondDRB,

                            v_DRB_ToAddModList,

                            v_SS_Drb_ConfigList,

                            v_QoS_Rules,

                            v_QoSFlowDescr);

    f_NR_AS_CipheringAlgorithm_Set(nea0);

    // Stop IP handling before reconfiguring SDAP to transparent mode

    f_IP_Handling_StopPDN(IP);

    // Reconfigure SDAP to transparent mode when moving to connected state

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(1, absent, absent, false),

                                                   v_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(1, absent, present_, true), // Removed mapping of QoS to default DRB

                                                   v_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_SecondDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_ConfigTransparentMode), //Configure SDAP in transparent mode

                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_FirstDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_ConfigTransparentMode)}; //Configure SDAP in transparent mode

    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,

                                     TEST_LOOPModeB_ON,

                                     -,-,-,-,-,

                                     v_DRB_ToAddModList,

                                     v_SS_Drb_ConfigList);

    f_NR_TestBody_Set(true);

    fl_TC_7_1_4_2_NR_TestBody(v_FirstDRB, v_SecondDRB);

    f_NR_TestBody_Set(false);

    //Postamble

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

    f_NR_ReleaseAllCells();

  }

After change

   function f_TC_7_1_4_2_NR5GC() runs on NR5GC_PTC

  {//SDAP Data Transfer handling without Header UL/DL

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;

    var IntegerList_Type v_DRBList;

    var integer v_FirstDRB;

    var integer v_SecondDRB;

    var template (value) QoS_Rules v_QoS_Rules;

    var template (value) QoSFlowDescr v_QoSFlowDescr;

    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var PDN_Index_Type v_PDN_Index;

    var template (value) QosFlow_Identification_Type v_QosFlowId;

    var integer v_PduSessionId; 

    f_NR5GC_Init_NAS(NR_1);

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    //Preamble

    // Perform Registration and get PDU Establishment Request message

    v_GSM_MobilityInfo_Type := fl_SDAP_Preamble_Part1 (nr_Cell1);

    //Get DRB identities

    v_FirstDRB := tsc_NR_DRB1;

v_SecondDRB := tsc_NR_DRB2;

    v_PduSessionId := oct2int(v_GSM_MobilityInfo_Type.SessionId.sessionId);
    //Note: In order to be able to start the IP handling during the preamble, QFI=2 is mapped temporarily to the default DRB (v_FirtDRB).

    //      This mapping will be removed when moving to connected mode

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_SecondDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_Configuration(v_PduSessionId, Absent, Absent)), //Configure SDAP in normal mode for the moment

                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_FirstDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_Configuration(v_PduSessionId, Absent, Present, {2}))}; //Configure SDAP in normal mode for the moment

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(

                                                   cs_SDAP_Config_Sdap(v_PduSessionId, absent, absent, false),

                                                   v_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(

                                                   cs_SDAP_Config_Sdap(v_PduSessionId, absent, present_, true, {2}),

                                                   v_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_QoS_Rules := cs_QoS_Rules ({fl_BuildQoSRuleRemoteAccess("4"), fl_BuildQoSRuleRemoteAccess("5")});

    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow2, cs_QoS_Flow3});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,

    // then stop it before the test body, so that TTCN can route the IP data

    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);

    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId.sessionId), 2); //Indicate the QoS flow associated with the default DRB to start the IP handling

    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));

    fl_SDAP_Preamble_Part2 (nr_Cell1,

                            v_GSM_MobilityInfo_Type,

                            v_FirstDRB,

                            v_SecondDRB,

                            v_DRB_ToAddModList,

                            v_SS_Drb_ConfigList,

                            v_QoS_Rules,

                            v_QoSFlowDescr);

    f_NR_AS_CipheringAlgorithm_Set(nea0);

    // Stop IP handling before reconfiguring SDAP to transparent mode

    f_IP_Handling_StopPDN(IP);

    // Reconfigure SDAP to transparent mode when moving to connected state

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(

                                                    cs_SDAP_Config_Sdap(v_PduSessionId, absent, absent, false),

                                                   v_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(

                                                    cs_SDAP_Config_Sdap(v_PduSessionId, absent, present_, true), // Removed mapping of QoS to default DRB

                                                   v_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_SecondDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_ConfigTransparentMode(Absent)), //Configure SDAP in transparent mode

                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,

                                                             v_FirstDRB,

                                                             -, //RLC normal mode

                                                             -, //PDCP normal mode

                                                             -,

                                                             cs_SDAP_ConfigTransparentMode(Present))}; //Configure SDAP in transparent mode

    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,

                                     TEST_LOOPModeB_ON,

                                     -,-,

                                      -,

                                     -,-,

                                     v_DRB_ToAddModList,

                                     v_SS_Drb_ConfigList);

    f_NR_TestBody_Set(true);

    fl_TC_7_1_4_2_NR_TestBody(v_FirstDRB, v_SecondDRB, v_PduSessionId);

    f_NR_TestBody_Set(false);

    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);
    //Postamble

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

    f_NR_ReleaseAllCells();

  }

Change 2

	Function name
	fl_TC_7_1_4_2_NR_TestBody

	Reason for change
	1. The PDU session ID used in SDAP configuration should reflect the value received from the UE IN PDU SESSION ESTABLISHMENT REQUEST
2. The current implementation of the parallel behaviour at 6 does not work as the test case will proceed as soon as either expected input is recieved, causing the other one to be unexpected when it is received and fail the test case. 
3. At step 8 the received SDAP data PDU on DRB 2 will have no UL SDAP header

4. The RRCReconfiguration message sent at step 5 should include two entries in DrbToAddModList according to 38.523-1 7.1.4.2.3.2-1



	Summary of change
	1. Use PDU session ID from the established PDU session instead of ‘1’

2. Modify step 6 to wait for RRCConnectionReconfigurationComplete in parallel with SDAP end-marker PDU
3. Modify expected PDU format at step 8
4. Add entry for DRB 1 to DrbToAddModList
Note : An associated prose CR on 38.523-1 will be raised at RAN5#85



	TTCN module
	SDAP_NR5GC

	MCC160 Comment
	1. Accepted, conditional to prose CR approval.
2. Accepted but implemented differently (see below).

3. Accepted.

4. Accepted, but it should be {} instead of omit (see below).


Before change

      function fl_TC_7_1_4_2_NR_TestBody(integer p_FirstDRB, integer p_SecondDRB) runs on NR_BASE_PTC

  {

    var template (present) SDAP_PDU_Type v_ExpectedPdu;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3

    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4

    var default v_DefaultRef;

    // @siclog "Step 1" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI =2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 2" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=2?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2), v_IPDataQFI2);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 1 for QFI 5.

    f_SDAP_SendNoHeader(nr_Cell1, p_FirstDRB, v_IPDataQFI5);

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=5?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5), v_IPDataQFI5);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // The SS transmits an NR RRCReconfiguration message to configure QoS Flow rules

    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(1, absent, absent, false, {2}),

                                                   p_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,

                                                         v_DRB_ToAddModList,

                                                         omit,

                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));

    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,

                                      tsc_NR_RbId_SRB1,

                                      cs_TimingInfo_Now,

                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig))));

    // EXCEPTION: In parallel to the event described in step 8 the events specified in Table 7.1.4.2.3.2-2 shall take place.

    // @siclog "Table 7.1.4.2.3.2-2, Step 1" siclog@

    // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=2?

    v_DefaultRef := activate(a_SDAP_RxEndMarker(nr_Cell1, p_FirstDRB, tsc_QFI2));
    // @siclog "Step 6" siclog@

    // The UE transmits an RRCReconfigurationComplete message.

    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    // @siclog "Step 7" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI=2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 8" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 2?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2), v_IPDataQFI2);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    deactivate(v_DefaultRef);
  …..

  }

}

After change

 
function fl_TC_7_1_4_2_NR_TestBody(integer p_FirstDRB, integer p_SecondDRB, integer p_PduSessionId) runs on NR_BASE_PTC 

  {

    var template (present) SDAP_PDU_Type v_ExpectedPdu;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3

    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4

    var default v_DefaultRef;

    var boolean v_Step6 := false;

    var boolean v_RxEndMark := false;
    // @siclog "Step 1" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI =2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 2" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=2?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2), v_IPDataQFI2);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 1 for QFI 5.

    f_SDAP_SendNoHeader(nr_Cell1, p_FirstDRB, v_IPDataQFI5);

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=5?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5), v_IPDataQFI5);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // The SS transmits an NR RRCReconfiguration message to configure QoS Flow rules

    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(

                                                   cs_SDAP_Config_Sdap(p_PduSessionId, absent, absent, false, {2}),

                                                   p_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(

                                                   cs_SDAP_Config_Sdap(p_PduSessionId, absent, present_, true),

                                                   p_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,

                                                         v_DRB_ToAddModList,

                                                         omit,

                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));

    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,

                                      tsc_NR_RbId_SRB1,

                                      cs_TimingInfo_Now,

                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig))));

    // @siclog "Step 6" siclog@

    // The UE transmits an RRCReconfigurationComplete message.

    // EXCEPTION: In parallel to the event described in step 8 the events specified in Table 7.1.4.2.3.2-2 shall take place.

    // @siclog "Table 7.1.4.2.3.2-2, Step 1" siclog@

    // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=2?

    alt {

      [not v_Step6] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))

        {

            v_Step6 := true;

            if (not v_RxEndMark){

              repeat;

              }

        }

      [not v_RxEndMark]  DRB.receive (car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI2)}))

        {

          f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");

          v_RxEndMark := true;

           if (not v_Step6){

              repeat;

              }

        }

    };
    // @siclog "Step 7" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI=2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 8" siclog@

// Check: Does the UE transmits SDAP Data PDU on DRB 2? 

    v_ExpectedPdu := cr_SDAP_PDU_UL_NoHeader(v_IPDataQFI2);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");

   ….

  }

}

MCC160 proposed implementation for changes 2.2 and 2.4
  function fl_TC_7_1_4_2_NR_TestBody(integer p_FirstDRB, integer p_SecondDRB, integer p_PduSessionId) runs on NR_BASE_PTC //@sic R5s190948 sic@

  {

    var template (present) SDAP_PDU_Type v_ExpectedPdu;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3

    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4

    var boolean v_EndMarker := false; // @sic R5s190948 sic@; boolean to indicate if the End-Marker has been received

    var boolean v_RrcMsg := false; // @sic R5s190948 sic@; boolean to indicate if the RRC message has been received
    // @siclog "Step 1" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI =2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 2" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=2?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2), v_IPDataQFI2);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 1 for QFI 5.

    f_SDAP_SendNoHeader(nr_Cell1, p_FirstDRB, v_IPDataQFI5);

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, which is default DRB, with SDAP header including QFI=5?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5), v_IPDataQFI5);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // The SS transmits an NR RRCReconfiguration message to configure QoS Flow rules

    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS

    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(p_PduSessionId, absent, absent, false, {2}),

                                                   p_SecondDRB,

                                                   cs_38508_PDCP_Config(ms300)),

                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(p_PduSessionId, absent, present_, true, {}), //@sic R5s190948 sic@

                                                   p_FirstDRB,

                                                   cs_38508_PDCP_Config(ms300))};

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,

                                                         v_DRB_ToAddModList,

                                                         omit,

                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));

    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,

                                      tsc_NR_RbId_SRB1,

                                      cs_TimingInfo_Now,

                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig))));

    // EXCEPTION: In parallel to the event described in step 8 the events specified in Table 7.1.4.2.3.2-2 shall take place.

    // @siclog "Table 7.1.4.2.3.2-2, Step 1" siclog@ 

    // @siclog "Step 6" siclog@

    //@sic R5s190948 sic@

    do {

      alt {

        [not v_RrcMsg] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))

          { // The UE transmits an RRCReconfigurationComplete message.

            v_RrcMsg := true;

            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

          }

        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI2)}))

          { // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=2? 

            v_EndMarker := true;

            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");

          }

      }

    }

    while (not(v_RrcMsg and v_EndMarker));   

    // @siclog "Step 7" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 2 for QFI=2.

    f_SDAP_SendNoHeader(nr_Cell1, p_SecondDRB, v_IPDataQFI2);

    // @siclog "Step 8" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 2?

    v_ExpectedPdu := cr_SDAP_PDU_UL_NoHeader(v_IPDataQFI2); //@sic R5s190948 sic@

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9" siclog@

    // The SS sends the SDAP Data PDU without SDAP header on DRB 1 for QFI=5.

    f_SDAP_SendNoHeader(nr_Cell1, p_FirstDRB, v_IPDataQFI5);

    // @siclog "Step 10" siclog@

    // Check: Does the UE transmits SDAP Data PDU on DRB 1, with SDAP header including QFI=5?

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5), v_IPDataQFI5);

    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {v_ExpectedPdu}));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  }

5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR5GC band n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_7_1_4_2_LOG.html
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

